Supplementary Information (SI) for Physical Chemistry Chemical Physics.


# NMR methods for characterizing molecular species within two immiscible solvents: application to SABRE-hyperpolarised species

Guillaume Verhaeghe,[a] Gaspard Huber [a] and Patrick Berthault*[a]

## SUPPORTING INFORMATION




**Figure S1**: $M_{xy}$ profile of a REBURP pulse of 1 ms calibrated to provide a 180° flip angle on-resonance. Simulation using trajectories on a Bloch sphere for 7 spins resonating at offsets -4035, -2690, -1345, 0, 1345, 2690, 4035 Hz. Blue and yellow background colours are added to simulate where could be the water and organic phases when the pulse is used simultaneously with a field gradient and the interface is set on-resonance as in OSDS.




**Figure S2:** a) Simulation of the slice selection scheme. Slice order given by the viridis scale. (Simulation parameters: matrix size = 500x500 pixels, circle radius = 245 pixels, slice width = 30 pixels. These parameters are in accordance with those of the experiment), b) Plot showing the percentage of pixels already present in the previous slices. Transposed to the Radsl-CSI, this shows that for most slices, ca. 30% of the spins have already been excited. An optimised order ($\theta$ = {0°, 90°, 45°, 135°…}) can also be chosen in order to decrease this loss.




**Figure S3**: Time evolution of the acetonitrile $^{1}H$ NMR signal in each pair of z-slices of the biphasic sample (organic phase). Slice 3 represents the lowest slice usable for the $CD_2Cl_2$ phase, slice 17 is at the interface with the aqueous phase (the two lowest pairs of z-slices have been discarded). Distance between two pairs of slices: $\Delta z$ = 0.064 cm. The curves consist of 278 datapoints, separated by a time interval of 8 min 33 s. The fit curves are superimposed.




**Figure S4**: Time evolution of the acetonitrile $^1$H NMR signal in each pair of z-slices of the biphasic sample (aqueous phase). Slice 30 is the highest usable for the $D_2O$ phase (pairs of slices 31 and 32 have been discarded). Distance between two pairs of z-slices: $\Delta z$ = 0.064 cm.




**Figure S5:** Pulse sequence of the 3D CSI-DOSY experiment. Direct dimension (F3): $^{1}$H spectra; first indirect dimension (F1): spatial localization; second indirect dimension (F2): self-diffusion. On the rf channel, the narrow rectangles represent $\pi/2$ hard pulses, while the wider rectangles represent $\pi$ pulses. For the CSI part, the phase gradient is applied along z. Bipolar diffusion gradients of duration $\delta$ and of double sine shape (in green) are employed along y, in stimulated echo mode to avoid too much $T_2$ relaxation. After the second $\pi/2$ pulse, when magnetization should only be on z, a purge gradient at the magic angle ($G_x = G_y = G_z$) is used (displayed in red). For the imaging part: $\tau_1$ = 1.45 ms, phase gradient duration 1.25 ms, F1 phase dimension made of 64 points from $-G_{max}$ to $G_{max} - \Delta G$, with $G_{max}$ = 2.43 G.cm$^{-1}$ and the gradient increment $\Delta G$ = 0.08 G.cm$^{-1}$. For the self-diffusion measurement: $\tau_2$ = 0.7 ms, $\Delta$ = 200 ms, $\delta$ = 1 ms, linear ramp of the diffusion gradients constituted by 16 nominal values from 1.79 to 25.13 G.cm$^{-1}$.

```
; 3D CSI-DOSY
; BRUKER avance II-version
; written for a z-profile image (using a spin echo)
; and for the measurement of the diffusion in the y-direction
; (using a stimulated echo together with bipolar gradients)

#include <Avance.incl>
#include <Delay.incl>
#include <De.incl>
#include <Grad.incl>

define list<gradient> diff=<Difframp>            ; list containing the y-gradient values

"DELTA=(d20-p1*2-p2-p30*2-d16*3-p19)"            ; big Delta
"p2=2*p1"

lgrad r2d<2d> = td1
aqseq 312

1 ze
2 d8
3 d1 UNBLKGRAD
4 4u pl1:f1                                      ; set pulse power on f1, the proton channel
  p1 ph1                                         ; exc. pulse
  50u
  d14
  d15
  p2 ph2
  d15
  d14 grad{(0)|(0)|(0)+r2d(cnst23)}              ; phase-encoding along z
  50u groff
  p30:gp6*diff                                   ; diffusion
```

```
  d16
  p2 ph4
  p30:gp6*-1*diff
  d16
  p1 ph3
  p19:gp7                                 ; purge gradient
  d16
  DELTA
  p1 ph5
  p30:gp6*diff
  d16
  p2 ph4
  p30:gp6*-1*diff
  d16
  5u BLKGRAD
  go=3 ph31
  1m igrad r2d
  d11 wr #0 if #0 zd
  lo to 3 times td1                       ; F1 (spatial) dimension loop
  1m zgrad r2d
  1m igrad diff
  lo to 2 times td2                       ; F2 (diffusion) dimension loop
exit

ph1=0
ph2=0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
ph3=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph4=0 0 0 0 2 2 2 2
ph5=0 1 2 3
ph31=0 3 2 1 0 3 2 1 2 1 0 3 2 1 0 3

; gp6: bipolar gradient (for instance, gpy6 = 100)
; pl1: power level for high power pulse
; p1: 90 degrees high power pulse
; p2: 180 degrees high power pulse
; p19: duration of the purge gradient
; p30: duration of bipolar diffusion gradients
; d14: duration of the phase encoding gradient
; d15: echo delay
; d16: delay for gradient recovery
; cnst23: constant for the phase z-gradient
```




**Figure S6:** Self-diffusion curves extracted for acetonitrile in the different z-slices (12-31 range) of the sample. Between two slices $\Delta z = 0.039$ cm. The logarithm of the ratio $S_0/S$ is plotted as a function of $\beta = (\gamma\delta G)^2(\Delta - \delta/3)$, and the best-fit line is superimposed. Here S is the signal area for a given gradient value, while $S_0$ is the signal area for the minimal gradient used.




**Figure S7:** Self-diffusion curves extracted for acetonitrile in the different z-slices (32-33 and 37-54 range) of the sample. Between two slices $\Delta z$ = 0.039 cm. The logarithm of the ratio $S_0/S$ is plotted as a function of $\beta = (\gamma\delta G)^2(\Delta - \delta/3)$, and the best-fit line is superimposed. Here S is the signal area for a given gradient value, while $S_0$ is the signal area for the minimal gradient used.




**Figure S8:** a) 3D-contour plot showing the signal spreading as a function of the y-gradient strength in percentage for different z coordinates; b) extracted y-gradient strength as a function of the position, expressed here as a function of the slice number (1, bottom of the tube; 64, top of the tube). The results are shown for acetonitrile, c) without and d) with the 2D corrected y-gradient strengths. The profile of the diffusion coefficient D is successfully flattened in each phase.




**Figure S9.** Details of the RadsI-CSI experiments shown in Figure 7 of the main text. a) Hyperpolarised and b) thermal contour plots, on which two magnitude mode spectra are extracted (bottom: in the organic phase; top: in the aqueous phase. On the hyperpolarised spectra, the stars indicate a residual signal arising from the other phase (insufficient spatial resolution). The data are in magnitude mode with subtraction of the first row.

```
; RadslCSI
; tested on AvanceII
; 2D-sequence (spectral – spatial)
; gradient-echo, z-profile, longitudinal slices

#include <Avance.incl>
#include <Delay.incl>
#include <De.incl>
#include <Grad.incl>

"d14=p11/2"                                          ; can be modified to change the FOV
"cnst20=-cnst22*p11/(2*d14)"

lgrad cos = td1
lgrad sin = td1

define list<gradient> encoding = {0.0 -0.13 0.13 -0.25 0.25 -0.38 0.38 -0.5 0.5 -0.63 0.63 -0.75 0.75 -0.88 0.88 -1}
; normal = {-1.0 -0.88 -0.75 -0.63 -0.5 -0.38 -0.25 -0.13 0.0 0.13 0.25 0.38 0.5 0.63 0.75 0.88}
; partial = {0.0 0.13 0.25 0.38 0.5 0.63 0.75 0.88 1.0 1.13 1.25 1.38 1.5 1.63 1.75 1.88}
; shifted partial = {-0.25 -0.13 0.0 0.13 0.25 0.38 0.5 0.63 0.75 0.88 1.0 1.13 1.25 1.38 1.5 1.63}
; centric = {0.0 -0.13 0.13 -0.25 0.25 -0.38 0.38 -0.5 0.5 -0.63 0.63 -0.75 0.75 -0.88 0.88 -1}
; see (1)



1 ze
2 d1 UNBLKGRAD
4 4u pl0:f1
5 d12 grad{cos(cnst22)|sin(cnst22)|(0)}              ; slice gradient
6 p11:sp1:f1 ph1                                     ; soft excitation pulse (we use a Hamming-filtered sinc)
7 5u groff
  5u
  d14 grad{cos(cnst20)|sin(cnst20)|encoding(1)}      ; phase-encoding gradient (and refocusing of the slice gradient)
  5u groff
  5u BLKGRAD
  go=2 ph31
  1m igrad encoding
  1m igrad cos
  1m igrad sin
  d11 wr #0 if #0 zd
  lo to 2 times td1
exit

ph1=0 2 2 0 1 3 3 1
ph31=0 2 2 0 1 3 3 1

; pl0 : 0W
; spw1: f1 channel - shaped pulse power
; p11: f1 channel - excitation pulse length
; d14: encoding time
; cnst22: slice gradient strength (in % of the max. gradient)
; (1) encoding gradient: for the moment, has to be manually chosen between 'normal' ([-Gmax;+Gmax[), 'partial' [0;2Gmax[),
; 'shifted partial' (starts slightly before the center) or centric (0, +ΔG, -ΔG, etc., with ΔG the gradient increment).
; The lists are written such as Gmax = 1% with TD1 = 16, d14 = 1.25 ms and Gz = 60 G/cm. Then, FOV  = 2.5 cm.
```

**Figure S10.** Bruker pulse sequence of the Radsl-CSI, tested on a Avance II and Topspin 3.0.